# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=shellpop
pkgver=148.a145349
pkgrel=3
pkgdesc='Generate easy and sophisticated reverse or bind shell commands.'
arch=('any')
groups=('blackarch' 'blackarch-automation')
url='https://github.com/0x00-0x00/ShellPop'
license=('MIT')
depends=('python2' 'python2-pyperclip' 'python2-netifaces' 'python2-argcomplete')
makedepends=('git' 'python2-setuptools')
source=("$pkgname::git+https://github.com/0x00-0x00/ShellPop.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd $pkgname

  python2 setup.py build
}

package() {
  cd $pkgname

  python2 setup.py install --root="$pkgdir" --skip-build

  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md KNOWN_ISSUES
  install -Dm 644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

